# NAG Library Routine Document

# F07GAF (DPPSV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07GAF (DPPSV) computes the solution to a real system of linear equations

$$AX = B,$$

where $A$ is an $n$ by $n$ symmetric positive-definite matrix stored in packed format and $X$ and $B$ are $n$ by $r$ matrices.

## 2 Specification

```
SUBROUTINE F07GAF(UPLO, N, NRHS, AP, B, LDB, INFO)
INTEGER           N, NRHS, LDB, INFO
double precision  AP(*), B(LDB,*)
CHARACTER*1       UPLO
```

The routine may be called by its LAPACK name ***dppsv***.

## 3 Description

F07GAF (DPPSV) uses the Cholesky decomposition to factor $A$ as $A = U^{\mathrm{T}}U$ if UPLO = 'U' or $A = LL^{\mathrm{T}}$ if UPLO = 'L', where $U$ is an upper triangular matrix and $L$ is a lower triangular matrix. The factored form of $A$ is then used to solve the system of equations $AX = B$.

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphiahttp://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Parameters

1: UPLO – CHARACTER*1 *Input*

*On entry*: if UPLO = 'U', the upper triangle of $A$ is stored.

If UPLO = 'L', the lower triangle of $A$ is stored.

*Constraint*: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry*: $n$, the number of linear equations, i.e., the order of the matrix $A$.

*Constraint*: N $\geq$ 0.

3: NRHS – INTEGER *Input*

*On entry*: $r$, the number of right-hand sides, i.e., the number of columns of the matrix $B$.

*Constraint*: NRHS $\geq$ 0.

4: AP($*$) – ***double precision*** array *Input/Output*

**Note**: the dimension of the array AP must be at least $\max(1, \mathrm{N} \times (\mathrm{N}+1)/2)$.

*On entry*: the $n$ by $n$ symmetric matrix $A$, packed by columns.

More precisely,

if UPLO = 'U', the upper triangle of $A$ must be stored with element $A_{ij}$ in $\mathrm{AP}(i+j(j-1)/2)$ for $i \leq j$;

if UPLO = 'L', the lower triangle of $A$ must be stored with element $A_{ij}$ in $\mathrm{AP}(i+(2n-j)(j-1)/2)$ for $i \geq j$.

*On exit*: if INFO = 0, the factor $U$ or $L$ from the Cholesky factorization $A = U^{\mathrm{T}}U$ or $A = LL^{\mathrm{H}}$, in the same storage format as $A$.

5: B(LDB,$*$) – ***double precision*** array *Input/Output*

**Note**: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

*On entry*: the $n$ by $r$ right-hand side matrix $B$.

*On exit*: if INFO = 0, the $n$ by $r$ solution matrix $X$.

6: LDB – INTEGER *Input*

*On entry*: the first dimension of the array B as declared in the (sub)program from which F07GAF (DPPSV) is called.

*Constraint*: LDB $\geq \max(1, \mathrm{N})$.

7: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO $< 0$

If INFO $= -i$, the $i$th argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO $> 0$

If INFO $= i$, the leading minor of order $i$ of $A$ is not positive-definite, so the factorization could not be completed, and the solution has not been computed.

## 7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A+E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and $\epsilon$ is the ***machine precision***. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x}-x\|_1}{\|x\|_1} \leq \kappa(A)\frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \left\|A^{-1}\right\|_1 \|A\|_1$, the condition number of $A$ with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

F07GBF (DPPSVX) is a comprehensive LAPACK driver that returns forward and backward error bounds and an estimate of the condition number. Alternatively, F04BEF solves $Ax = b$ and returns a forward error bound and condition estimate. F04BEF calls F07GAF (DPPSV) to solve the equations.

## 8 Further Comments

The total number of floating-point operations is approximately $\frac{1}{3}n^3 + 2n^2r$, where $r$ is the number of right-hand sides.

The complex analogue of this routine is F07GNF (ZPPSV).

## 9 Example

This example solves the equations

$$Ax = b,$$

where $A$ is the symmetric positive-definite matrix

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 8.70 \\ -13.35 \\ 1.89 \\ -4.14 \end{pmatrix}.$$

Details of the Cholesky factorization of $A$ are also output.

### 9.1 Program Text

```
*     F07GAF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
*     .. Local Arrays ..
      DOUBLE PRECISION AP((NMAX*(NMAX+1))/2), B(NMAX)
*     .. External Subroutines ..
      EXTERNAL         DPPSV, X04CCF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07GAF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the upper or lower triangular part of the matrix A from
*        data file
*
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+(J*(J-1))/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+((2*N-J)*(J-1))/2),J=1,I),I=1,N)
         END IF
*
*        Read b from data file
*
         READ (NIN,*) (B(I),I=1,N)
*
*        Solve the equations Ax = b for x
*
         CALL DPPSV(UPLO,N,1,AP,B,N,INFO)
```

```
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Solution'
            WRITE (NOUT,99999) (B(I),I=1,N)
*
*           Print details of factorization
*
            WRITE (NOUT,*)
            IFAIL = 0
            CALL X04CCF(UPLO,'Non-unit diagonal',N,AP,'Cholesky factor',
     +                  IFAIL)
*
         ELSE
            WRITE (NOUT,99998) 'The leading minor of order ', INFO,
     +        ' is not positive definite'
         END IF
       ELSE
         WRITE (NOUT,*) 'NMAX too small'
       END IF
*
99999 FORMAT ((3X,7F11.4))
99998 FORMAT (1X,A,I3,A)
      END
```

## 9.2 Program Data

```
F07GAF Example Program Data
  4                           :Value of N
  4.16  -3.12   0.56  -0.10
         5.03  -0.83   1.18
                0.76   0.34
                       1.18 :End of matrix A
  8.70 -13.35   1.89  -4.14 :End of vector b
```

## 9.3 Program Results

```
 F07GAF Example Program Results

 Solution
        1.0000    -1.0000     2.0000    -3.0000

 Cholesky factor
             1          2          3          4
 1      2.0396    -1.5297     0.2746    -0.0490
 2                 1.6401    -0.2500     0.6737
 3                            0.7887     0.6617
 4                                       0.5347
```